Journal of Software for Algebra and Geometry

Setting the scene for Betti characters

Federico Galetto

vol 13 2023

JSAG **13** (2023), 45–51
https://doi.org/10.2140/jsag.2023.13.45

The Journal of Software for
Algebra and Geometry

Setting the scene for Betti characters

FEDERICO GALETTO

ABSTRACT: Finite group actions on free resolutions and modules arise naturally in many interesting examples. Understanding these actions amounts to describing the terms of a free resolution or the graded components of a module as group representations which, in the nonmodular case, are completely determined by their characters. With this goal in mind, we introduce a Macaulay2 package for computing characters of finite groups on free resolutions and graded components of finitely generated graded modules over polynomial rings.

1. INTRODUCTION. Let R be a polynomial ring over a field $\Bbbk$, and M be a finitely generated graded R-module. Let G be a linearly reductive group acting $\Bbbk$-linearly on R and M, and assume these actions preserve degrees and distribute over R-multiplication. If $F_\bullet$ is a minimal graded free resolution of M, then the action of G extends to $F_\bullet$. More precisely, if $\mathfrak{m}$ denotes the irrelevant maximal ideal of R, then the finite-dimensional vector spaces $F_i/\mathfrak{m}F_i$ carry a natural structure of graded G-representations (see [Galetto 2016, Proposition 2.4.9 and Remark 2.4.10] for details). This additional structure makes a resolution more rigid as the differentials must commute with the group action; in some cases, this makes it possible to construct the differentials explicitly using representation theory (see, for example, [Sam 2009; Sam and Weyman 2011]). Understanding how G acts on the modules $F_i/\mathfrak{m}F_i$ may also lead to interesting combinatorial descriptions of the Betti numbers of M, such as in [Galetto 2020, Corollary 4.12]. Free resolutions equipped with group actions (also known as equivariant resolutions) have found many important applications, such as the computation of Betti numbers of determinantal varieties [Lascoux 1978], and a proof of the existence of pure free resolutions [Eisenbud et al. 2011] (a central aspect of Boij-Söderberg theory).

From a computational perspective, the [Macaulay2] package `HighestWeights` [Galetto 2015] allows users to determine the representation theoretic structure of an equivariant resolution with the action of a semisimple Lie group in characteristic zero. Recent publications [Zamaere et al. 2014; Efremenko et al. 2018; Galetto et al. 2018; Bauer et al. 2019; Galetto 2020; Biermann et al. 2020; Murai 2020; Shibata and Yanagawa 2023; Raicu 2021; Murai and Raicu 2022] point to an interest in equivariant resolutions with actions of finite groups, particularly symmetric groups. However, at the time of writing, no software solution is available to compute such actions. The present article introduces the Macaulay2 package `BettiCharacters` to fill this gap. In the nonmodular case (i.e., when the characteristic of the field does

MSC2020: primary 13-04; secondary 13A50, 13D02, 13P20, 20C15.
Keywords: Macaulay2, equivariant resolution, finite group, Betti character.
`BettiCharacters` version 2.1

not divide the order of the group), finite-dimensional representations of finite groups are determined, up to isomorphism, by their characters (see [Serre 1977, Chapter 2] for an introduction to the subject). Thus understanding the G-action on a minimal free resolution $F_\bullet$ amounts to describing the graded characters of the representations $F_i/\mathfrak{m}F_i$ or, equivalently, the characters of the graded components $(F_i/\mathfrak{m}F_i)_j$. The uniqueness of minimal free resolutions implies $(F_i/\mathfrak{m}F_i)_j \cong \operatorname{Tor}_i^R(M, \Bbbk)_j$ as G-representations. Moreover, the character of G on $\operatorname{Tor}_i^R(M, \Bbbk)_j$ evaluated at the identity of G is the dimension of $\operatorname{Tor}_i^R(M, \Bbbk)_j$ as a $\Bbbk$-vector space, i.e., the (i, j)-th Betti number of M. Therefore we adopt the following definition, after which the package is named.

Definition. The *(i, j)-th Betti character of G on M*, denoted $\beta^G_{i,j}(M)$, is the character of G on $\operatorname{Tor}_i^R(M, \Bbbk)_j$.

The package `BettiCharacters` implements the algorithm described in [Galetto 2022, Algorithm 1], which in essence propagates the group action from the module M through a (previously computed) minimal free resolution $F_\bullet$. In addition, `BettiCharacters` also allows users to compute the characters of G on the graded components of M. The rest of this article illustrates the main functionalities of the package.[1] The author thanks the anonymous referee for carefully reviewing this work.

2. Example: a symmetric shifted ideal. Consider the ideal generated by all quadratic squarefree monomials in a ring with four variables.

```
i1 : R = QQ[x_1..x_4];

i2 : I = ideal apply(subsets(gens R,2),product)

o2 = ideal (x x , x x , x x , x x , x x , x x )
             1 2   1 3   2 3   1 4   2 4   3 4

o2 : Ideal of R

i3 : RI = res I

      1      6      8      3
o3 = R  <-- R  <-- R  <-- R  <-- 0
     0      1      2      3      4

o3 : ChainComplex
```

The symmetric group $\mathfrak{S}_4$ acts by permuting the ring variables and, in doing so, preserves the ideal. Thus the action passes to the quotient and its minimal free resolution. The equivariant structure of the resolution is described in [Galetto 2020, Theorem 4.11] and [Efremenko et al. 2018, Theorem 4.1]. The ideal also belongs to the larger class of symmetric shifted ideals, whose equivariant resolutions are described in [Biermann et al. 2020, Theorem 6.2]. To verify these results computationally, we first define the group action. Since characters are class functions (i.e., constant on conjugacy classes) it is enough to define a single group element per conjugacy class. In the case of a symmetric group, the method `symmetricGroupActors` returns the desired elements as one-row matrices of substitutions for the ring variables. Then we set up the action on the resolution as an object of type `ActionOnComplex` using the `action` method.

[1] All computations performed in Macaulay2 version 1.19.1 on Fedora 36, using `BettiCharacters` 2.0.

```
i4 : needsPackage "BettiCharacters";

i5 : S4 = symmetricGroupActors R

o5 = {| x_2 x_3 x_4 x_1 |, | x_2 x_3 x_1 x_4 |, | x_2 x_1 x_4 x_3 |,
     ---------------------------------------------------------------
      | x_2 x_1 x_3 x_4 |, | x_1 x_2 x_3 x_4 |}

o5 : List

i6 : A = action(RI,S4)

o6 = ChainComplex with 5 actors

o6 : ActionOnComplex
```

Now we can use the `character` method to compute Betti characters.

```
i7 : a = character A

o7 = Character over R

     (0, {0}) => | 1  1  1  1 1 |
     (1, {2}) => | 0  0  2  2 6 |
     (2, {3}) => | 0 -1  0  0 8 |
     (3, {4}) => | 1  0 -1 -1 3 |

o7 : Character
```

The output is of type `Character`, a new kind of hash table introduced by the package. The keys are pairs containing the homological degree and internal (multi)degree of the nonzero components of a graded character. The values are one-row matrices whose entries are the traces of the previously defined group elements.

Finally, we decompose this character against the character table of the symmetric group, which can be obtained using the method `symmetricGroupTable`. The irreducible characters χ^{λ} of $\mathfrak{S}_4$ are in bijection with the partitions λ of 4 (see [Fulton 1997, §7]), which appear as row labels in the character table. The column labels are the cardinalities of the conjugacy classes represented by the permutations in `o5`.

```
i8 : T = symmetricGroupTable R

o8 = Character table over R

             |   6   8   3   6  1
     --------+-------------------
        (4)  |   1   1   1   1  1
      (3,1)  |  -1   0  -1   1  3
         2   |
       (2 )  |   0  -1   2   0  2
         2   |
     (2,1 )  |   1   0  -1  -1  3
         4   |
       (1 )  |  -1   1   1  -1  1

o8 : CharacterTable
```

The decomposition is achieved with the method `decomposeCharacter`. The output is a table whose rows are labeled by pairs of homological degree and internal (multi)degree, and whose columns are labeled by irreducible characters of the group. The entry in a given cell is the coefficient of the irreducible character labeled by the column in the degree labeled by the row.

```
i9 : decomposeCharacter(a,T)

o9 = Decomposition table

                |                      2        2
                |  (4)  (3,1)  (2 )  (2,1 )
     -----------+-------------------------
     (0, {0})   |   1     0     0       0
     (1, {2})   |   1     1     1       0
     (2, {3})   |   0     1     1       1
     (3, {4})   |   0     0     0       1

o9 : CharacterDecomposition
```

The computation above shows that the Betti character $\beta_{1,2}^{\mathfrak{S}_4}$ of the quotient by the ideal in the example is $\chi^{(4)} + \chi^{(3,1)} + \chi^{(2,2)}$.

The methods in BettiCharacters are completely independent of the group. However, the package contains methods that simplify working with symmetric groups, as shown in the current example.

3. Example: Klein point configuration. We consider the Klein configuration of points in the projective plane. The defining ideal I is explicitly constructed in [Bauer et al. 2019, Proposition 7.3]. Although I is defined over the rationals, we work over the cyclotomic field obtained by adjoining a primitive seventh root of unity for the purpose of defining a group action.

```
i1 : kk=toField(QQ[a]/ideal(sum apply(7,i->a^i)));

i2 : R=kk[x,y,z];

i3 : f4=x^3*y+y^3*z+z^3*x

      3     3       3
o3 = x y + y z + x*z

o3 : R

i4 : f6=-1/54*det(jacobian transpose jacobian f4)

        5    5      2 2 2      5
o4 = x*y  + x z - 5x y z  + y*z

o4 : R

i5 : I=minors(2,jacobian matrix{{f4,f6}});

o5 : Ideal of R
```

The unique simple group G of order 168 acts on the projective plane preserving the Klein configuration. This induces an action on our polynomial ring preserving the ideal I. The action (which is minimally defined over our cyclotomic field) is explicitly described in [Bauer et al. 2019, §2.2]. In particular, the group is generated by elements g of order 7, h of order 3, and i of order 2. Since we are interested in some characters of G, we need a representative for each conjugacy class; therefore, in addition to g, h, and i, we also consider the identity element, the inverse of g, and an element j of order 4. We define all these group elements as matrices.

```
i6 : g=matrix{{a^4,0,0},{0,a^2,0},{0,0,a}};

                3        3
o6 : Matrix kk  <--- kk
```

```
i7 : h=matrix{{0,1,0},{0,0,1},{1,0,0}};

                3       3
o7 : Matrix ZZ  <--- ZZ

i8 : i=(2*a^4+2*a^2+2*a+1)/7 * matrix{{a-a^6,a^2-a^5,a^4-a^3},
                                       {a^2-a^5,a^4-a^3,a-a^6},
                                       {a^4-a^3,a-a^6,a^2-a^5}};

                3       3
o8 : Matrix kk  <--- kk

i9 : j=-1/(2*a^4+2*a^2+2*a+1) * matrix{{a^5-a^4,1-a^5,1-a^3},
                                        {1-a^5,a^6-a^2,1-a^6},
                                         {1-a^3,1-a^6,a^3-a}};

                3       3
o9 : Matrix kk  <--- kk

i10 : G={id_(R^3),i,h,j,g,inverse g};
```

As proved in [Seceleanu 2015, Theorem 4.4] and [Bauer et al. 2019, Proposition 8.1], the symbolic cube $I^{(3)}$ is not contained in the square I^2. The second proof of [Bauer et al. 2019, Proposition 8.1] reduces the failure of containment to showing the graded component of degree 21 in the quotient $I^{(2)}/I^2$ is a trivial G-representation. By local duality, this is equivalent to showing that the last module in a minimal free resolution of I^2 is generated in degree 24 by a one-dimensional trivial G-module. We proceed to compute the character of G on the last module of the resolution of I^2.

```
i11 : I2=I^2;

o11 : Ideal of R

i12 : RI2=res I2

       1      6      6      1
o12 = R  <-- R  <-- R  <-- R  <-- 0
      0      1      2      3      4

o12 : ChainComplex

i13 : needsPackage "BettiCharacters";

i14 : A=action(RI2,G,Sub=>false)

o14 = ChainComplex with 6 actors

o14 : ActionOnComplex
```

The action is defined with the option `Sub=>false`, which allows passing group elements as square matrices rather than one-row matrices of substitutions as in Section 2. Next we compute the character of the G-action on the resolution of I^2 in homological degree 3.

```
i15 : character(A,3)

o15 = Character over R

      (3, {24}) => | 1 1 1 1 1 1 |

o15 : Character
```

As expected, we obtain a trivial character concentrated in degree 24.

The `BettiCharacters` package can also compute the characters of a finite group on the graded components of a module. Using the package [SymbolicPowers], we can directly establish that the character of G on the graded component of degree 21 in $I^{(2)}/I^2$ is trivial.

```
i16 : needsPackage "SymbolicPowers";

i17 : Is2 = symbolicPower(I,2);

o17 : Ideal of R

i18 : M = Is2 / I2;

i19 : B = action(M,G,Sub=>false)

o19 = Module with 6 actors

o19 : ActionOnGradedModule

i20 : character(B,21)

o20 = Character over R

      (0, {21}) => | 1 1 1 1 1 1 |

o20 : Character
```

SUPPLEMENT. The online supplement contains version 2.1 of `BettiCharacters`.

REFERENCES.

[Bauer et al. 2019] T. Bauer, S. Di Rocco, B. Harbourne, J. Huizenga, A. Seceleanu, and T. Szemberg, “Negative curves on symmetric blowups of the projective plane, resurgences, and Waldschmidt constants”, *Int. Math. Res. Not.* **2019**:24 (2019), 7459–7514. MR

[Biermann et al. 2020] J. Biermann, H. de Alba, F. Galetto, S. Murai, U. Nagel, A. O’Keefe, T. Römer, and A. Seceleanu, “Betti numbers of symmetric shifted ideals”, *J. Algebra* **560** (2020), 312–342. MR

[Efremenko et al. 2018] K. Efremenko, J. M. Landsberg, H. Schenck, and J. Weyman, “On minimal free resolutions of sub-permanents and other ideals arising in complexity theory”, *J. Algebra* **503** (2018), 8–20. MR

[Eisenbud et al. 2011] D. Eisenbud, G. Flø ystad, and J. Weyman, “The existence of equivariant pure free resolutions”, *Ann. Inst. Fourier* (*Grenoble*) **61**:3 (2011), 905–926. MR Zbl

[Fulton 1997] W. Fulton, *Young tableaux*, London Mathematical Society Student Texts **35**, Cambridge University Press, 1997. MR Zbl

[Galetto 2015] F. Galetto, “Free resolutions and modules with a semisimple Lie group action”, *J. Softw. Algebra Geom.* **7** (2015), 17–29. MR Zbl

[Galetto 2016] F. Galetto, “Propagating weights of tori along free resolutions”, *J. Symbolic Comput.* **74** (2016), 1–45. MR Zbl

[Galetto 2020] F. Galetto, “On the ideal generated by all squarefree monomials of a given degree”, *J. Commut. Algebra* **12**:2 (2020), 199–215. MR Zbl

[Galetto 2022] F. Galetto, “Finite group characters on free resolutions”, *J. Symbolic Comput.* **113** (2022), 29–38. MR Zbl

[Galetto et al. 2018] F. Galetto, A. V. Geramita, and D. L. Wehlau, “Symmetric complete intersections”, *Comm. Algebra* **46**:5 (2018), 2194–2204. MR Zbl

[Lascoux 1978] A. Lascoux, “Syzygies des variétés déterminantales”, *Adv. in Math.* **30**:3 (1978), 202–237. MR Zbl

[Macaulay2] D. R. Grayson and M. E. Stillman, “Macaulay2, a software system for research in algebraic geometry”, software, available at http://www.math.uiuc.edu/Macaulay2/.

[Murai 2020] S. Murai, “Betti tables of monomial ideals fixed by permutations of the variables”, *Trans. Amer. Math. Soc.* **373**:10 (2020), 7087–7107. MR Zbl

[Murai and Raicu 2022] S. Murai and C. Raicu, "An equivariant Hochster's formula for $\mathfrak{S}_n$-invariant monomial ideals", *J. Lond. Math. Soc.* (2) **105**:3 (2022), 1974–2010. MR

[Raicu 2021] C. Raicu, "Regularity of $\mathfrak{S}_n$-invariant monomial ideals", *J. Combin. Theory Ser. A* **177** (2021), art. id. 105307. MR Zbl

[Sam 2009] S. V. Sam, "Computing inclusions of Schur modules", *J. Softw. Algebra Geom.* **1** (2009), 5–10. MR Zbl

[Sam and Weyman 2011] S. V. Sam and J. Weyman, "Pieri resolutions for classical groups", *J. Algebra* **329** (2011), 222–259. MR Zbl

[Seceleanu 2015] A. Seceleanu, "A homological criterion for the containment between symbolic and ordinary powers of some ideals of points in $\mathbb{P}^2$", *J. Pure Appl. Algebra* **219**:11 (2015), 4857–4871. MR Zbl

[Serre 1977] J.-P. Serre, *Linear representations of finite groups*, Graduate Texts in Mathematics **42**, Springer, 1977. MR Zbl

[Shibata and Yanagawa 2023] K. Shibata and K. Yanagawa, "Elementary construction of minimal free resolutions of the Specht ideals of shapes $(n-2, 2)$ and $(d, d, 1)$", *J. Algebra Appl.* **22**:9 (2023), Paper No. 2350199, 26. MR

[SymbolicPowers] E. Grifo, "SymbolicPowers", Macaulay2 package, available at https://github.com/Macaulay2/M2/tree/master/M2/Macaulay2/packages.

[Zamaere et al. 2014] C. B. Zamaere, S. Griffeth, and S. V. Sam, "Jack polynomials as fractional quantum Hall states and the Betti numbers of the $(k+1)$-equals ideal", *Comm. Math. Phys.* **330**:1 (2014), 415–434. MR Zbl

RECEIVED: 30 Jun 2021 REVISED: 26 Feb 2023 ACCEPTED: 30 May 2023

FEDERICO GALETTO:

f.galetto@csuohio.edu
Department of Mathematics and Statistics, Cleveland State University, Cleveland, OH, United States